Optimal scattering of 3 points into the unit square.

```
> x[1],y[1] := (0,0); y[2] := 1; x[3] := 1;
```

$$x_1, y_1 := 0, 0$$

$$y_2 := 1$$

$$x_3 := 1 \quad (1)$$

```
> eqns := {(1-y[3])^2+(1-x[2])^2=m^2, y[3]^2+1=m^2, x[2]^2+1=m^2};
```

$$eqns := \{x_2^2 + 1 = m^2, y_3^2 + 1 = m^2, (1 - y_3)^2 + (1 - x_2)^2 = m^2\} \quad (2)$$

```
> F := [seq( lhs(e)-rhs(e), e=eqns ) ];
```

$$F := [x_2^2 + 1 - m^2, y_3^2 + 1 - m^2, (1 - y_3)^2 + (1 - x_2)^2 - m^2] \quad (3)$$

```
> G := Groebner[Basis]( F, plex(x[2],y[3],m) );
```

$$G := [16\, m^2 - 16\, m^4 + m^6, -m^4 + 4\, m^2\, y_3, y_3^2 + 1 - m^2, -m^2 + 2\, y_3 + 2\, x_2] \quad (4)$$

```
> M := G[1];
```

$$M := 16\, m^2 - 16\, m^4 + m^6 \quad (5)$$

```
> factor(M);
```

$$m^2\, (16 - 16\, m^2 + m^4) \quad (6)$$

```
> fsolve(M=0,m);
```

$$-3.86370330515627, -1.03527618041008, 0., 0., 1.03527618041008, 3.86370330515627 \quad (7)$$

```
> solutions := solve( m^4-16*m^2+16, m );
```

$$solutions := \sqrt{6} - \sqrt{2}, -\sqrt{6} + \sqrt{2}, \sqrt{6} + \sqrt{2}, -\sqrt{6} - \sqrt{2} \quad (8)$$

```
> evalf(solutions);
```

$$1.035276181, -1.035276181, 3.863703305, -3.863703305 \quad (9)$$

```
> eval(eqns,m=0);
```

$$\{x_2^2 + 1 = 0, y_3^2 + 1 = 0, (1 - y_3)^2 + (1 - x_2)^2 = 0\} \quad (10)$$

The solutions for m=0 are complex. Let's discard those

```
> Groebner[Basis]([op(F),1-m*z],plex(x[2],y[3],z,m));
```

$$[16 - 16\, m^2 + m^4, -16\, m + m^3 + 16\, z, -m^2 + 4\, y_3, -m^2 + 4\, x_2] \quad (11)$$

Since we're only interested in computing m, this is the more efficient variable ordering to use to do that, i.e., compute $I \cap \mathbf{Q}[m]$.

```
> Groebner[Basis]([op(F),1-m*z],lexdeg([x[2],y[3],z],[m]));
```

$$[16 - 16\, m^2 + m^4, -16\, m + m^3 + 16\, z, -m^2 + 4\, y_3, -m^2 + 4\, x_2] \quad (12)$$

Optimal scattering of 6 points into the unit square.

```
> restart;
> eqns := {x[6]=1/2, x[6]^2+y[6]^2=m^2,
          x[4]=1/2, x[4]^2+(1-y[3])^2=m^2,
          x[6]^2+(y[3]-y[6])^2=m^2};
```

$$eqns := \left\{x_4 = \frac{1}{2}, x_6 = \frac{1}{2}, x_4^2 + (1-y_3)^2 = m^2, x_6^2 + y_6^2 = m^2, x_6^2 + (y_3 - y_6)^2 = m^2\right\} \quad \textbf{(13)}$$

```
> F := [seq( lhs(e)-rhs(e), e=eqns )];
```

$$F := \left[x_4 - \frac{1}{2}, x_6 - \frac{1}{2}, x_4^2 + (1-y_3)^2 - m^2, x_6^2 + y_6^2 - m^2, x_6^2 + (y_3 - y_6)^2 - m^2\right] \quad \textbf{(14)}$$

```
> G := Groebner[Basis]( [op(F)], lexdeg([x[6],y[6],x[4],y[3]],[m]));
```

$$G := [144\,m^4 + 65 - 232\,m^2, 12\,y_3\,m^2 + 10 - 8\,m^2 - 15\,y_3, 2\,x_4 - 1, 12\,y_6\,m^2 + 5 - 4\,m^2 - 15\,y_6, 2\,x_6 - 1, 4\,y_3^2 + 5 - 4\,m^2 - 8\,y_3, 8\,y_3\,y_6 + 5 - 4\,m^2 - 8\,y_3, 4\,y_6^2 + 1 - 4\,m^2] \quad \textbf{(15)}$$

```
> factor(G[1]);
```

$$(-5 + 4\,m^2)\,(36\,m^2 - 13) \quad \textbf{(16)}$$

```
> fsolve(4*m^2-5,m);
```

$$-1.118033989, 1.118033989 \quad \textbf{(17)}$$

```
> fsolve(36*m^2-13,m);
```

$$-0.6009252126, 0.6009252126 \quad \textbf{(18)}$$

There seem to be two possibilities. If you look at the picture you might see that m cannot be +1.1 and hence it must be 0.60. But, let's consider each case separately.

```
> G1 := Groebner[Basis]( [op(F),36*m^2-13],
     lexdeg([x[6],y[6],x[4],y[3]],[m]));
```

$$G1 := [36\,m^2 - 13, 3\,y_3 - 2, 2\,x_4 - 1, 3\,y_6 - 1, 2\,x_6 - 1] \quad \textbf{(19)}$$

```
> G2 := Groebner[Basis]( [op(F),4*m^2-5,y[3]],
     lexdeg([x[6],y[6],x[4],y[3]],[m]));
```

$$G2 := [-5 + 4\,m^2, y_3, 2\,x_4 - 1, 2\,x_6 - 1, y_6^2 - 1] \quad \textbf{(20)}$$

The case $4\,m^2 - 5$ leads to $y_3 = 0$ and $y_6 = 1$ so the other case must be the right one. Why did we get this case anyway? Because there is no restriction on circle 3 not being on top of circle 2. One way to eliminate this case is

```
> G1 := Groebner[Basis]( [op(F),1-(1-y[6])*y[3]*z],
     lexdeg([x[6],y[6],x[4],y[3],z],[m]));
```

$$G1 := [36\,m^2 - 13, 4\,z - 9, 3\,y_3 - 2, 2\,x_4 - 1, 3\,y_6 - 1, 2\,x_6 - 1] \quad \textbf{(21)}$$

The other case $36\,m^2 = 13$.

```
> solve(G1[1],m);
```

$$\frac{1}{6}\sqrt{13}, -\frac{1}{6}\sqrt{13} \quad \textbf{(22)}$$

```
> _EnvExplicit := true: solve( {op(G1),m=sqrt(13)/6} );
```

(23)

$$\left\{m = \frac{1}{6}\sqrt{13},\ z = \frac{9}{4},\ x_4 = \frac{1}{2},\ x_6 = \frac{1}{2},\ y_3 = \frac{2}{3},\ y_6 = \frac{1}{3}\right\} \tag{23}$$